Fourier Series Examples
(c) 2012 Damon A. Miller
damon.miller@wmich.edu
15 May 2012

References:
1. Weisstein, Eric W. “Fourier Series.” From MathWorld--A Wolfram Web Resource. http://mathworld.wolfram.com/-FourierSeries.html
2. J. A. Cadzow and H. F. Van Landingham, Signals, Systems, and Transforms, Prentice-Hall, Inc., New Jersey, 1985.
3. C. K. Alexander and M. N. O. Sadiku, Fundamentals of Electric Circuits, 4th ed., McGraw-Hill, Boston, 2009.

Square Wave

```
f[t_] := If[t <= 1 / 2, 1, 0]; T = 1; ω0 = 2 Pi / T;

Plot[f[t], {t, 0, T}, Frame -> True]
```




```
a0 = 1 / T Integrate[f[t], {t, 0, T}]
```

$$\frac{1}{2}$$

```
a[n_] = 2 / T Integrate[f[t] Cos[n ω0 t], {t, 0, T}]
```

$$\frac{\text{Sin}[n\,\pi]}{n\,\pi}$$

```
b[n_] = 2 / T Integrate[f[t] Sin[n ω0 t], {t, 0, T}]
```

$$\frac{2\,\text{Sin}\left[\frac{n\,\pi}{2}\right]^2}{n\,\pi}$$

```
fhat[t_, N_] := a0 + Sum[a[n] Cos[n ω0 t] + b[n] Sin[n ω0 t], {n, 1, N}]
```

```
Plot[{f[t], fhat[t, 1], fhat[t, 2], fhat[t, 10]},
 {t, 0, T}, PlotRange -> All, Frame -> True]
```




```
fhat[t, 5]
```

$$\frac{1}{2}+\frac{2\,\text{Sin}[2\,\pi\,t]}{\pi}+\frac{2\,\text{Sin}[6\,\pi\,t]}{3\,\pi}+\frac{2\,\text{Sin}[10\,\pi\,t]}{5\,\pi}$$

- **Estimate Low Pass RC Filter Response to a Square Wave**

```
H[s_] := 1 / (s + 1)
```

```
Table[{ω, Abs[H[I ω]], N[Arg[H[I ω]] / Degree]}, {ω, 1, 5, 2}] // TableForm
```

1	$\frac{1}{\sqrt{2}}$	-45.
3	$\frac{1}{\sqrt{10}}$	-71.5651
5	$\frac{1}{\sqrt{26}}$	-78.6901

```
vo[t_, N_] := Abs[H[I 0]] a0 + Sum[Abs[H[I n ω0]] a[n] Cos[n ω0 t + Arg[H[I n ω0]]] +
    Abs[H[I n ω0]] b[n] Sin[n ω0 t + Arg[H[I n ω0]]], {n, 1, N}]
```

```
vo[t, 5]
```

$$\frac{1}{2}+\frac{2\,\text{Sin}[2\,\pi\,t-\text{ArcTan}[2\,\pi]]}{\pi\sqrt{1+4\,\pi^2}}+\frac{2\,\text{Sin}[6\,\pi\,t-\text{ArcTan}[6\,\pi]]}{3\,\pi\sqrt{1+36\,\pi^2}}+$$

$$(2\,\text{Sin}[10\,\pi\,t-\text{ArcTan}[10\,\pi]])\Big/\left(5\,\pi\sqrt{1+100\,\pi^2}\right)$$

```
Plot[{f[t], fhat[t, 10], vo[t, 10]}, {t, 0, T}, Frame -> True]
```



Here is the output obtained used LTspice (linear.com). Note that edges are attenuated but the filter "passes" the 1/2V DC component.





- **Estimate High Pass RC Filter Response to a Square Wave**

```
H[s_] := s / (s + 1)
```

```
Table[{ω, Abs[H[I ω]], N[Arg[H[I ω]] / Degree]}, {ω, 1, 5, 2}] // TableForm
```

1	$\frac{1}{\sqrt{2}}$	45.
3	$\frac{3}{\sqrt{10}}$	18.4349
5	$\frac{5}{\sqrt{26}}$	11.3099

```
vo[t_, N_] := Abs[H[I 0]] a0 + Sum[Abs[H[I n ω0]] a[n] Cos[n ω0 t + Arg[H[I n ω0]]] +
    Abs[H[I n ω0]] b[n] Sin[n ω0 t + Arg[H[I n ω0]]], {n, 1, N}]
```

```
vo[t, 5]
```

$$\frac{4\,\text{Sin}\left[10\,\pi\,t+\text{ArcTan}\left[\frac{1}{10\,\pi}\right]\right]}{\sqrt{1+100\,\pi^2}}+\frac{4\,\text{Sin}\left[6\,\pi\,t+\text{ArcTan}\left[\frac{1}{6\,\pi}\right]\right]}{\sqrt{1+36\,\pi^2}}+\frac{4\,\text{Sin}\left[2\,\pi\,t+\text{ArcTan}\left[\frac{1}{2\,\pi}\right]\right]}{\sqrt{1+4\,\pi^2}}$$

```
Plot[{f[t], fhat[t, 20], vo[t, 10], vo[t, 40]}, {t, 0, T}, Frame -> True]
```



Here is the output obtained used LTspice (linear.com). Note that the system "passes" the edges and attenuates the constant portions of the input. Also note that the truncated Fourier Series does not fully pass the edges since the truncated higher frequency components are needed to represent the edges.



■ Sawtooth Wave

```
f[t_] := t; T = 1; ω0 = 2 Pi / T;
```

```
Plot[f[t], {t, 0, T}, Frame → True]
```




```
a0 = 1 / T Integrate[f[t], {t, 0, T}]
```

$$\frac{1}{2}$$

```
a[n_] = 2 / T Integrate[f[t] Cos[n ω0 t], {t, 0, T}]
```

$$\frac{1}{2\,n^2\,\pi^2}\,(-1+\text{Cos}[2\,n\,\pi]+2\,n\,\pi\,\text{Sin}[2\,n\,\pi])$$

```
b[n_] = 2 / T Integrate[f[t] Sin[n ω0 t], {t, 0, T}]
```

$$\frac{1}{2\,n^2\,\pi^2}\,(-2\,n\,\pi\,\text{Cos}[2\,n\,\pi]+\text{Sin}[2\,n\,\pi])$$

```
fhat[t_, N_] := a0 + Sum[a[n] Cos[n ω0 t] + b[n] Sin[n ω0 t], {n, 1, N}]
```

```
Clear[t]; Plot[{f[t], fhat[t, 1], fhat[t, 2], fhat[t, 10], fhat[t, 100]},
 {t, 0, T}, PlotRange → All, Frame → True]
```




```
fhat[t, 5]
```

$$\frac{1}{2}-\frac{\text{Sin}[2\,\pi\,t]}{\pi}-\frac{\text{Sin}[4\,\pi\,t]}{2\,\pi}-\frac{\text{Sin}[6\,\pi\,t]}{3\,\pi}-\frac{\text{Sin}[8\,\pi\,t]}{4\,\pi}-\frac{\text{Sin}[10\,\pi\,t]}{5\,\pi}$$